

## Practice: Scoring Using the SCORE Statement in PROC GLMSELECT

You want to re-create the model that was built in the previous practice (based on **stat1.ameshousing3**), create an item store, and then use the item store to score the new cases in **stat1.ameshousing4**. You'll score the data in two ways (using PROC GLMSELECT and PROC PLM) and compare the results.

1. Open the solution program from the previous practice, **st106s01.sas**. There is no need to examine the results, so make the following changes to the code:
  - a. Remove the PLOTS= option.
  - b. Add the NOPRINT option to the PROC GLMSELECT statement.
  - c. Remove the TITLE statement

Here's the modified code:

```
proc glmselect data=STAT1.ameshousing3
    seed=8675309
    noprint;
    class &categorical / param=ref ref=first;
    model SalePrice=&categorical &interval /
        selection=stepwise
        (select=aic
        choose=validate) hierarchy=single;
    partition fraction(validate=0.3333);
run;
```

2. In the PROC GLMSELECT step,
  - a. Add a STORE statement to create an item store named **store1**, and a SCORE statement to score the data in **stat1.ameshousing4**.
  - b. Add a PROC PLM step that uses the item store, **store1**, to score the data in **stat1.ameshousing4**.
 

**Note:** Be sure to use different names for the two scored data sets.
  - c. Add a PROC COMPARE step to compare the scoring results from PROC GLMSELECT and PROC PLM.
  - d. Submit the code and examine the results.

```
/*st106s02.sas*/

proc glmselect data=STAT1.ameshousing3
    seed=8675309
    noprint;
    class &categorical / param=ref ref=first;
    model SalePrice=&categorical &interval /
        selection=stepwise
        (select=aic
        choose=validate) hierarchy=single;
    partition fraction(validate=0.3333);
    score data=STAT1.ameshousing4 out=score1;
    store out=store1;
    title "Selecting the Best Model using Honest Assessment";
run;

proc plm restore=store1;
    score data=STAT1.ameshousing4 out=score2;
run;

proc compare base=score1 compare=score2 criterion=0.0001;
    var P_SalePrice;
```

```
with Predicted;  
run;
```

Here are the [results](#).

3. Does the PROC COMPARE output indicate any differences between the predictions produced by the two scoring methods?

The two scoring methods produce the same predictions.

**Note:** Depending on the version of SAS and SAS/STAT that you are using, your results might look somewhat different from the output shown here. However, the results should indicate that these data sets do not differ.

Hide Solution