Print

Practice: Using PROC GLM to Perform Post Hoc Pairwise Comparisons

Consider the analysis of the **garlic** data set. In the previous exercise, you used PROC GLM to perform one-way ANOVA, and found that there was a statistically significant difference among mean garlic bulb weights for the different fertilizers. Now, perform a post hoc test to look at the individual differences among means.

1. Use PROC GLM to conduct pairwise comparisons with an experimentwise error rate of α=0.05. (Use the Tukey adjustment.) Submit the code and view the results.

```
/*st102s02.sas*/
ods graphics;

ods select lsmeans diff diffplot controlplot;
proc glm data=STAT1.Garlic
         plots(only)=(diffplot(center) controlplot);
   class Fertilizer;
   model BulbWt=Fertilizer;
   Tukey: lsmeans Fertilizer / pdiff=all adjust=tukey;
   title "Post-Hoc Analysis of ANOVA - Fertilizer as Predictor";
run;
quit;

title;
```

   Here are the results.

2. Which types of fertilizer are significantly different?

   The Tukey comparisons show significant differences between fertilizers 3 and 4 (p=0.0020) and 1 and 4 (p=0.0058).

3. Use level 4 (the chemical fertilizer) as the control group and perform a Dunnett's comparison with the organic fertilizers to see whether they affected the average weights of garlic bulbs differently from the control fertilizer.

```
/*st102s02.sas*/
ods graphics;

ods select lsmeans diff diffplot controlplot;
proc glm data=STAT1.Garlic
         plots(only)=(diffplot(center) controlplot);
   class Fertilizer;
   model BulbWt=Fertilizer;
   Dunnett:lsmeans Fertilizer / pdiff=control('4') adjust=dunnett;
   title "Post-Hoc Analysis of ANOVA - Fertilizer as Predictor";
run;
quit;

title;
```

   Here are the results.

4. Which types of fertilizer are significantly different?

   The Dunnett comparisons show the same pairs as significantly different, but with smaller *p*-values than with the Tukey comparisons (3 versus 4 p=0.0011, 1 versus 4 p=0.0031). This is due to the fact that the Tukey adjustment is for more pairwise comparisons than the Dunnett adjustment.

5. **Challenge**: Perform unadjusted tests of all pairwise comparisons to see what would happen if the multi-test adjustments were not made.

```
/*st102s02.sas*/
ods graphics;

ods select lsmeans diff diffplot controlplot;
proc glm data=STAT1.Garlic
         plots(only)=(diffplot(center) controlplot);
   class Fertilizer;
   model BulbWt=Fertilizer;
   No_Adjust: lsmeans Fertilizer / pdiff=all adjust=t;
   title "Post-Hoc Analysis of ANOVA - Fertilizer as Predictor";
run;
quit;

title;
```

   Here are the results.

6. How do the results compare to what you saw in the Tukey adjusted tests?

   The unadjusted (t test) comparisons have smaller *p*-values than they had with Tukey adjustments. One additional comparison has a *p*-value below 0.05 (2 versus 3).

Hide Solution