

Practice: Performing a Two-Way ANOVA Using PROC GLM

Data were collected to determine whether different dosage levels of a drug have an effect on blood pressure for people with one of three types of heart disease. The data are in the **stat1.drug** data set.

1. Examine the data with a vertical line plot. Put **BloodP** on the Y axis, and **DrugDose** on the X axis, and then stratify by **Disease**.

```
/*st103s01.sas*/ /*Part A*/
proc sgplot data=STAT1.drug;
  vline DrugDose / group=Disease stat=mean response=BloodP markers;
  format DrugDose dosefmt.;
run;
```

Here are the [results](#).

2. What information can you obtain by looking at the data?

It seems that the drug dose affects a change in blood pressure. However, that effect is not consistent across diseases. Higher doses result in increased blood pressure for patients with disease B, decreased blood pressure for patients with disease A, and little change in blood pressure for patients with disease C.

3. Test the hypothesis that the means are equal. Be sure to include an interaction term if the graphical analysis that you performed indicates that would be advisable.

```
/*st103s01.sas*/ /*Part B*/
ods graphics on;
proc glm data=STAT1.drug plots(only)=intplot;
  class DrugDose Disease;
  model BloodP=DrugDose|Disease;
  lsmeans DrugDose*Disease;
run;
quit;
```

Here are the [results](#).

4. What conclusions can you reach at this point?

The global F test indicates a significant difference among the different groups. Because the interaction is in the model, this is a test of all combinations of **DrugDose*Disease** against all other combinations. The R-square value implies that approximately 35% of the variation in **BloodP** can be explained by variations in the explanatory variables. The interaction term is statistically significant, as predicted by the plot of the means.

5. To investigate the interaction effect between the two factors, include the SLICE option by manually editing the generated code or you can write the code directly.

Modify the LSMEANS statement to include the *slice=Disease* option preceded by a slash, as shown in the code below.

```
/*st103s01.sas*/ /*Part B*/  
ods graphics on;  
proc glm data=STAT1.drug plots(only)=intplot;  
  class DrugDose Disease;  
  model BloodP=DrugDose|Disease;  
  lsmeans DrugDose*Disease / slice=Disease;  
run;  
quit;
```

Here are the [results](#).

6. Is the effect of **DrugDose** significant?

The slice table shows the effect of **DrugDose** at each level of the disease. The effect is significant for all, except Disease C.

Hide Solution