

Using the LSMESTIMATE Statement to Estimate Relationships of Interest

In this demonstration, we generate an analysis of variance that tests and estimates the previously discussed contrasts.

Because PROC GLM is run with a STORE statement, you do not view the results now but use them for postfitting analyses. The STORE statement requests that PROC GLM save the context and results of the statistical analysis for use with PROC PLM.

The RESTORE option in the PROC PLM statement specifies the source item store for processing. This option is required because, in contrast to SAS data sets, there is no default item store. An item-store specification consists of a one- or two-level name as with SAS data sets. As with data sets, the default library association of an item store is with the Work library, and any stores created in this library are deleted when the SAS session concludes.

PROC PLM also contains the LSMESTIMATE statement, which provides a mechanism for obtaining custom hypothesis tests among least squares means. The LSMESTIMATE statement has several options. The DIVISOR option specifies a value by which to divide all coefficients so that fractional coefficients can be entered as integers. The ELSM option requests that the K matrix coefficients be displayed. These are the coefficients that apply to the L-S means. This option is useful to ensure that you assigned the coefficients correctly to the L-S means. Let's submit the code.

```
ods select none;
proc glm data=mydata.school;
  class school gender;
  model reading3 = school|gender;
  store out=mydata.schoolstore;
run;
quit;
ods select all;
proc plm restore=mydata.schoolstore;
  lsestimate school*gender 'Female Cottonwood&Dogwood vs. Female
                          Maple&Pine'
                          .5 0 .5 0 -.5 0 -.5 0 / elsm;
  lsestimate school 'Cottonwood vs. Dogwood, Maple and Pine'
                  1 -0.333333 -0.333333 -0.333333;
  lsestimate school 'Cottonwood vs. Dogwood, Maple and Pine'
                  3 -1 -1 -1 / divisor=3;
run;
```

The Store Information table and Class Level Information table are produced by default for the stored GLM model.

The ELSM option in the LSMESTIMATE statement generates the following output. This estimate is designed to compute the difference between the female students' average test scores at Cottonwood and Dogwood and the average of female students' test scores at Maple and Pine. The output shows the order of the least squares means and enables you to see whether you created the intended estimate.

The LSMESTIMATE provides the estimate, its standard error and degrees of freedom, a t-statistic, and a *p*-value. The first LSMESTIMATE statement indicates that the average of the scores of the female students of Cottonwood and Dogwood is 24.75 points lower than the average of the scores of the female students at Maple and Pine. This difference is significantly different from 0 because the *p*-value is 0.0056.

The last two LSMESTIMATE statements produced identical results. The difference is in the syntax. One statement used the decimal coefficients, and the other one used the integer coefficients with the DIVISOR= option. The results show that the **Reading3** test scores for Cottonwood are 23.54 points below the average **Reading3** test scores for Dogwood, Maple, and Pine. This difference is significantly different from zero because the associated *p*-value = 0.0009.

Close